

Querying RDF Streams with C-SPARQL *

Davide Francesco Barbieri

Daniele Braga

Stefano Ceri

Emanuele Della Valle

Michael Grossniklaus

Politecnico di Milano, Dipartimento di Elettronica e Informazione
Piazza L. da Vinci 32, 20133 Milano, Italy

firstname.lastname@elet.polimi.it

ABSTRACT

Continuous SPARQL (C-SPARQL) is a new language for continuous queries over streams of RDF data. C-SPARQL queries consider windows, i.e., the most recent triples of such streams, observed while data is continuously flowing. Supporting streams in RDF format guarantees interoperability and opens up important applications, in which reasoners can deal with evolving knowledge over time. Examples of such application domains include real-time reasoning over sensors, urban computing, or social semantic data. In this paper, we present the C-SPARQL language extensions both in terms of syntax and examples. Finally, we discuss existing applications that already use C-SPARQL and give an outlook on future research opportunities.

1. INTRODUCTION

Stream-based data sources such as sensors, feeds, click streams, and stock quotations have become increasingly important in many application domains. Streaming data are received continuously and in real-time, either implicitly ordered by arrival time, or explicitly associated with timestamps. As it is typically impossible to store a stream in its entirety, Data Stream Management Systems (DSMS) [14], e.g., [19, 12, 4, 1, 3], allow one to register continuously running queries that return new results as new data flow within the streams [15]. At the same time, reasoning upon very large RDF data collections is advancing fast, and SPARQL [24] has gained the role of standard query language for RDF data. Also, SPARQL engines are now capable of querying integrated repositories and collecting data from multiple sources. Still, the large knowledge bases now accessible via SPARQL (such as Linked Life Data¹) are static, and knowledge evolution is not adequately supported.

The combination of static RDF data with streaming information leads to **stream reasoning** [13], an important step enabling logical reasoning in real time on gigantic and inevitably noisy data streams in order to support the decision process of extremely large numbers of concurrent users. So far, this step has received little attention by the Semantic Web community. C-SPARQL is an extension of SPARQL designed to express continuous queries, i.e., queries registered over both RDF repositories and *RDF streams*. C-SPARQL queries can be considered as inputs to specialized

reasoners that use their knowledge about a domain to make real-time decisions. In such applications, reasoners operate upon knowledge snapshots, which are continuously refreshed by continuous queries. It is important to note that, in this view, reasoners can be unaware of time changes and of the existence of streams. Reasoning over streaming facts is also addressed by the authors of [30]. Their work is complementary to ours, as they focus on the scalability of reasoning techniques. Another research related to ours is that by Law et al. [17], who put particular emphasis on the problem of mining data streams [18]

C-SPARQL extends SPARQL with functionality required to manage streams, in a way that is comparable to the approach taken by CQL [2]. A previous attempt [10] to extend SPARQL to support data streams can be considered an antecedent of our work, as it introduced a syntax for the specification of logical and physical windows in SPARQL queries by means of local grammar extensions. However, their approach is different from ours at least in two key aspects. First, they simply introduce RDF streams as a new data type, and omit essential ingredients, such as aggregates and timestamp functions. Second, the authors do not follow the established approach where windows are used to transform streaming data into non-streaming data in order to apply standard algebraic operations. Instead, Bolles et al. have chosen to change the standard SPARQL operators by making them timestamp-aware and, thereby, actually introduce a new language semantics.

In the setting of stream processing, the capability to compute aggregations is an important functionality. When we started working on C-SPARQL, we based it on SPARQL 1.0 which does not contain any support for aggregates. In previous publications, we therefore also introduced our own syntax and semantics for aggregates in C-SPARQL that does not shrink results in the presence of grouping [7]. In the meantime, it is foreseeable that the upcoming SPARQL 1.1 specification will include aggregation functionality similar to the one known from SQL. For this paper and future work on C-SPARQL, we have chosen to align our notion of aggregates with the one proposed by the W3C and present all examples accordingly.

Nevertheless, we would like to point out that several implementations of SPARQL support some form of proprietary aggregation functions and group definitions. OpenLink Virtuoso² supports COUNT, COUNT DISTINCT, MAX, MIN and AVG, with implicit grouping criteria. ARQ³ supports COUNT and COUNT

*This work is supported by the European project LarKC (FP7-215535). Michael Grossniklaus's contribution was carried out under the SNF grant number PBEZ2-121230.

¹<http://www.linkedlifedata.com/>

²<http://virtuoso.openlinksw.com/>

³<http://jena.sourceforge.net/ARQ/>

DISTINCT over groups defined through an SQL-like GROUP BY clause. ARC⁴ also supports the keyword AS to bind variables to aggregated results. In [26], the authors study how aggregation and grouping can be defined in the context of queries over RDF graphs, taking into consideration the peculiarities of the data model, and providing an extension of SPARQL based on operational semantics.

This paper is organized as follows. In Section 2, we present the distinguishing language extensions of C-SPARQL referring to a very simple scenario of social data analysis. After introducing the RDF stream data type, we discuss the extensions for windows, stream registration, and query registration. Other application scenarios, beyond social data analysis, are presented in Section 3. Finally, outlooks about using C-SPARQL for enabling stream reasoning are presented in Section 4. Note that this paper does not discuss neither the evaluation and optimization of C-SPARQL queries, nor other entailment regimes beyond basic RDF entailment. For further information about these topics, the reader is directed to [7] and [6].

2. C-SPARQL

We present C-SPARQL by progressively introducing its new features relative to SPARQL. We interleave the presentation of the new syntax, extended by adding new productions to the standard grammar of SPARQL [24], and the discussion of some examples of usage. As a demonstration scenario, we have chosen queries that are relevant to a (highly simplified) case of social data analysis.

2.1 RDF Stream Data Type

C-SPARQL adds **RDF streams** to the data types supported by SPARQL, in the form of an extension done much in the same way in which the stream type has been introduced to extend relations in relational data stream management systems. An RDF stream is defined as an ordered sequence of pairs, where each pair is made of an RDF triple and its timestamp τ :

$$\begin{aligned} & \dots \\ & (\langle subj_i, pred_i, obj_i \rangle, \tau_i) \\ & (\langle subj_{i+1}, pred_{i+1}, obj_{i+1} \rangle, \tau_{i+1}) \\ & \dots \end{aligned}$$

Timestamps can be considered as *annotations* of RDF triples, and are monotonically non-decreasing in the stream ($\tau_i \leq \tau_{i+1}$). More precisely, timestamps are not strictly increasing because they are not required to be unique. Any (unbounded, though finite) number of consecutive triples can have the same timestamp, meaning that they “occur” at the same time, although sequenced in the stream according to some positional order.

Example. We refer to a very simple scenario of social data analysis. The entities and relationships that we consider are described in the schema of Figure 1. All entities are identified by URLs.

Users also have names, and, by virtue of two cyclic relationships, *know* and *follow* other users. Using well-known Semantic Web vocabularies, the user name and the knows relationship can be described using Friend of a Friend vocabulary (FOAF) [9]. For the “knows” relationship, we can use Semantically-Interlinked Online Communities (SIOC) [8].

Objects represent entities of the real world (such as movies or books, to give examples that are relevant for our sce-

⁴<http://arc.semsol.org/>




Figure 1: Data model

nario), with an optional name and a mandatory category. *Resources* represent sources of information that *describe* the actual objects, such as Web pages about a particular book or movie. As vocabularies for describing objects and resources, we can use `rdfs:label` [11] for the name and Simple Knowledge Organization System (SKOS) [23] to link an object to its category, by means of the *subject* attribute. Moreover, categories identifiers are assumed to be taken from YAGO [28].

All the information described so far is static (or, more precisely, slowly changing), meaning that the applications we are willing to consider can assume this information as invariant in a period comparable with the size of a window. Of course updates of this information are also allowed, e.g., to state that a new friendship holds by virtue of the addition of an instance of the *knows* relationship.

The running example also shows streaming information, and namely streams of notifications that capture the behavior of users with respect to resources (and therefore, transitively, to objects). The *accesses*, *likes*, and *dislikes* relationships represent events that occur at the time in which users access a resource or express their opinion about it.⁵ Quite straightforwardly, any interactions of a user with a resource generates in the stream a triple of the form $\langle U, sd:accesses, R \rangle$, where U and R respectively represent a generic user and a generic resource. Also, selected interactions of users with resources generate triples of the form $\langle U, sd:likes, R \rangle$ and $\langle U, sd:dislikes, R \rangle$. It is worth noting that in the stream the predicates can only assume one of the three values exemplified above, while the subjects and objects may freely vary in the space of users and resources. This is coherent with RDF repositories whose predicates are taken from a small vocabulary constituting a sort of schema. However, the interpretation of C-SPARQL makes no specific assumptions nor requires restrictions on variable bindings relative to any part of the streaming triples. An example of possible triples in a stream of interactions and opinions is given below.

triple	Timestamp
c:Usr1 sd:accesses c:movie1	t_{400}
c:Usr2 sd:accesses c:movie1	t_{401}
c:Usr1 sd:likes c:movie2	t_{402}
...	...

⁵In the rest of this paper, we refer to this vocabulary with the prefix *sd*.

2.2 Windows

The introduction of data streams in C-SPARQL requires the ability to *identify* such data sources and to specify *selection* criteria over them.

As for *identification*, we assume that each data stream is associated with a distinct IRI, that is a locator of the actual data source of the stream. More specifically, the IRI represents an IP address and a port for accessing streaming data. As for *selection*, given that streams are intrinsically infinite, we introduce the notion of windows upon streams, whose types and characteristics are inspired by those defined for relational streaming data.

Identification and selection are expressed in C-SPARQL by means of the `FROM STREAM` clause. The syntax is as follows:

```
FromStrClause → 'FROM' ['NAMED'] 'STREAM' StreamIRI
               '[ RANGE' Window ']'
Window         → LogicalWindow | PhysicalWindow
LogicalWindow → Number TimeUnit WindowOverlap
TimeUnit      → 'ms' | 's' | 'm' | 'h' | 'd'
WindowOverlap → 'STEP' Number TimeUnit | 'TUMBLING'
PhysicalWindow → 'TRIPLES' Number
```

A window extracts the *last* data elements from the stream, which are the only part of the stream to be considered by one execution of the query. Such extraction can be *physical* (a given number of triples) or *logical* (all triples occurring within a given time interval, whose number is variable over time).

Logical windows are *sliding* [16] when they are progressively advanced of a given `STEP` (i.e., a time interval that is shorter than the window's time interval). They are *non-overlapping* (or `TUMBLING`) when they are advanced of exactly their time interval at each iteration. With tumbling windows every triple of the stream is included exactly into one window, whereas with sliding windows some triples can be included into several windows.

The optional `NAMED` keyword works exactly like when applied to the standard SPARQL `FROM` clause for tracking the provenance of triples. It binds the IRI of a stream to a variable which is later accessible through the `GRAPH` clause.

Example. As a very simple first example, consider the query that extracts all books (i.e., all objects whose category is “book”) seen by the friends of John in the last 15 minutes. The query considers the last 15 minutes, and the sliding window is modified every minute, so that the query result is renewed every minute.

```
SELECT DISTINCT ?object
FROM STREAM <http://streamingsocialdata.org/
            interact.trdf> [RANGE 15m STEP 1m]
WHERE { ?user sd:accesses ?resource .
        ?user foaf:knows ?jhn .
        ?jhn foaf:name "John" .
        ?resource t:describes ?object .
        ?object skos:subject yago:Movies . }
```

The query joins static and streaming knowledge, and is executed as follows. First, all triples with `sd:accesses` as a predicate are extracted from the current window over the stream, to match the first triple pattern in the `WHERE` clause. Then the other triple patterns are matched against the static knowledge base, applying the “join” conditions expressed by the bindings of variables `?user` and `?resource` to identify the observed `?objects`. The window considers all the stream triples in the last 15 minutes, and is advanced every minute.

This means that at every new minute new triples enter into the window and old triples exit from the window. Note that the query result does not change during the slide interval, and is only updated at every slide change. Triples arriving in the stream between these points in time are queued until the next slide change and do not contribute to the result until then.

2.3 Stream Registration

The result of a C-SPARQL query can be a set of bindings, but also a new RDF stream. In order to generate a stream, the query must be registered through the following statement:

```
Registration → 'REGISTER STREAM' QueryName
               ['COMPUTED EVERY' Number TimeUnit] 'AS' Query
```

Only queries in the `CONSTRUCT` and `DESCRIBE` form⁶ can be registered as generators of RDF streams, as they produce RDF triples, associated with a timestamp as an effect of the query execution.

The optional `COMPUTED EVERY` clause indicates the frequency at which the query *should* be computed. If no frequency is specified, the query is computed at a frequency that is automatically determined by the system.⁷

Example. The following example shows the construction of a new RDF data stream by means of the registration of a `CONSTRUCT` query. We consider the previous example again, and modify it so as to generate a stream by selecting all interactions that are of the “likes” type, that are performed by a friend of John, and that concern movies.

```
REGISTER STREAM MoviesJohnsFriendsLike
COMPUTED EVERY 5m AS
CONSTRUCT {?user sd:likes ?resource}
FROM STREAM <http://streamingsocialdata.org/
            interact.trdf> [RANGE 30m STEP 5m]
WHERE { ?user sd:likes ?resource .
        ?user foaf:knows ?john .
        ?john foaf:name "John" .
        ?resource sd:describes ?object .
        ?object skos:subject yago:Movies . }
```

This query uses the same logical conditions as the previous one on static data, but only matches the `sd:likes` predicate. The output is constructed in the format of a stream of RDF triples. Every query execution may produce from a minimum of zero triples to a maximum of an entire graph. The timestamp is always dependent on the query execution time only, and is not taken from the triples that match the patterns in the `WHERE` clause. Thus, even though in the example the output stream is a restriction of the input stream, a new timestamp is assigned to every triple. Also note that, if the window contains more than one matching triple with a `sd:likes` predicate, then also the result contains more than

⁶There are four query forms in SPARQL, different in the first clause: `SELECT` returns variables bound in a query pattern match. `CONSTRUCT` returns an RDF graph constructed by substituting variables in a set of triple templates. `ASK` returns a boolean indicating whether a query pattern matches or not. `DESCRIBE` returns an RDF graph that describes the resources found. Please refer to [24] for further explanations.

⁷Several data stream management systems are capable of self tuning the execution frequency of registered queries. This not only applies to queries with unspecified registration frequencies, but also whenever, due to peaks of workload, the execution frequency of all queries is reduced, so as to gracefully degrade the overall performances.

one triple, that are returned as a graph. In this case the same timestamp is assigned to all the triples of the graph. In all cases, however, timestamps are system-generated in monotonic non-decreasing order. Results of two evaluations of the previous query are presented in the table below, generating two graphs (one at $\tau = 100$ and one at $\tau = 101$).

triple	Timestamp
c:Usr1 sd:likes c:Movie1	t_{100}
c:Usr2 sd:likes c:Movie2	t_{100}
c:Usr1 sd:likes c:Movie2	t_{101}
c:Usr2 sd:likes c:Movie1	t_{101}
c:Usr3 sd:likes c:Movie3	t_{101}

2.4 Query Registration

All queries over RDF data streams are denoted as *continuous queries*, because they continuously produce output in the form of tables of variable bindings or RDF graphs. In the section above we addressed the registration of RDF streams. Here, we address the registration of queries that do not produce a stream, but a result that is periodically updated. C-SPARQL queries are registered through the following statement:

```
Registration  $\rightarrow$  'REGISTER QUERY' QueryName
                ['COMPUTED EVERY' Number TimeUnit] 'AS' Query
```

The `COMPUTED EVERY` clause is the same as the one for stream registration.

Example. As a very simple example of a registered query that does not generate a stream, consider the following query. For each known user, the query counts the overall number of interactions performed in the last 30 minutes and the number of distinct objects to which the resources refer.

```
REGISTER QUERY GlobalCountOfInteractions
                COMPUTED EVERY 5m AS
SELECT ?user
      COUNT(?resource) as ?numberOfInteractions
      COUNT(DISTINCT ?obj) as ?numDifferentObjects
FROM STREAM <http://streamingsocialdata.org/
             interact.trdf> [RANGE 30m STEP 5m]
WHERE { ?user sd:accesses ?resource .
        ?resource sd:describes ?obj . }
GROUP BY { ?user }
```

The query is executed by matching all interactions in the window, grouping them by `?user`, and computing the aggregates. The result has the form of a table of bindings that is updated every 5 minutes.

All the examples considered so far has shown a join of static and streaming knowledge. We now show a simple queries that only takes a stream as input, and namely the stream generated by the query shown in Section 2.3.

```
REGISTER QUERY GlobalCountOfInteractions
                COMPUTED EVERY 5m AS
SELECT ?user COUNT(?resource) as ?numberOfMovies
FROM STREAM <http://streamingsocialdata.org/
             MoviesJohnsFriendsLike.trdf> [RANGE 30m STEP 5m]
WHERE { ?user sd:likes ?resource }
GROUP BY { ?user }
```

The query counts, among the friends of John, the number of movies that each friend has liked in the last 30 minutes.

2.5 Multiple Streams

C-SPARQL queries can combine triples from more than

one RDF stream, as shown in the next example.

Example. We now consider, in addition to the stream of interactions, the presence of a second stream of data concerning the entrance of registered users into theaters to watch movies. The next queries takes as input the stream of preferences of John's friends and the stream about people entering cinemas, and identifies friends who like a 3D movies, but only considering users who actually watched at least two 3D movies in the last week (so as to focus on the advice of "experts").

```
REGISTER QUERY JohnsFriendsToRecommend3DMovies AS
SELECT ?user
FROM STREAM <http://streamingsocialdata.org/
             MoviesJohnsFriendsLike.trdf> [RANGE 1h]
FROM STREAM <http://comingsoon.com/
             WatchedMovies.trdf> [RANGE 7d]
WHERE {
  ?user sd:likes ?resource .
  ?resource sd:describes ?object .
  ?object skos:subject yago:3DMovies .
  { SELECT ?user
    WHERE {
      ?user sd:watches ?res1 .
      ?res1 sd:describes ?object1 .
      ?object1 skos:subject yago:3DMovies . }
    GROUP BY ?user
    HAVING COUNT(DISTINCT ?object1) >= 2 }
}
```

The query is executed as follows. Variable `?user` is matched in the `WHERE` clause of the outer query among the friends of John. Also, the `object` is checked to be a 3D movie (the stream is selected checking that the objects are classified as generic movies). The user is also checked to have the property of having seen at least two other 3D movies in the nested query. Note the use of the same `?user` variable in the nested query so as to pass the binding and check the "aggregate" property.

2.6 Timestamp Function

The timestamp of a stream element can be retrieved and bound to a variable using a timestamp function. The timestamp function has two arguments.

- The first is the name of a variable, introduced in the `WHERE` clause and bound by pattern matching with an RDF triple of that stream.
- The second (optional) is the URI of a stream, that can be obtained through `SPARQL GRAPH` clause.

The function returns the timestamp of the RDF stream element producing the binding. If the variable is not bound, the function is undefined, and any comparison involving its evaluation has a non-determined behavior. If the variable gets bound multiple times, the function returns the most recent timestamp value relative to the query evaluation time.

Example. In order to exemplify the use of timestamps within queries, we show a query that tries to discover causal relationships between different actions. More precisely, the query identifies users who are likely to influence the behavior of other users, by matching interactions of the same kind that occur on the same resource *after* the first user has performed them. The query in C-SPARQL is the following:

```
REGISTER STREAM OpinionMakers
                COMPUTED EVERY 5m AS
SELECT ?opinionMaker
FROM STREAM <http://streamingsocialdata.org/
```

```

interact.trdf> [RANGE 30m STEP 5m]
WHERE { ?opinionMaker foaf:knows ?friend .
        ?friend ?opinion ?resource .
        ?opinionMaker ?opinion ?resource .
        FILTER ( timestamp(?friend) >
                 timestamp(?opinionMaker)
                 && ?opinion != sd:accesses ) }
GROUP BY ( ?opinionMaker )
HAVING ( COUNT(DISTINCT ?friend) > 3 )

```

Note that the timestamps are taken from variables that occur only once in patterns applied to streaming triples, thus avoiding ambiguity. Also, the query filters out actions of type “accesses”, that are normally required before expressing an opinion such as “like” or “dislike”.

3. APPLICATIONS

The scenario of social data analysis is just one example of many possible applications of C-SPARQL. In the last years, more and more effort has been put in trying to address problems that require reasoning on streaming data, and this has been done mainly with “classical” reasoning tools. For instance, Bandini et al. [5] worked on traffic monitoring and traffic pattern detection. Mendler et al. [22] applied constructive Description Logics to financial-transaction auditing. In the mobile telecommunication sector, Luther et al. [20] reported the need for reasoning over streams for situation-aware mobile services. Walavalkar et al. [30] worked on patient monitoring systems. All these application areas are natural settings for C-SPARQL. In the following, we provide more details about concrete applications of C-SPARQL in the cases of situation aware mobility and oil production. In Section 4 we will also outline how we are currently studying dedicated reasoning techniques for the interplay of C-SPARQL and reasoners, in order to efficiently carry out reasoning tasks over streams.

3.1 Situation-Aware Mobility

Mobility is one of the defining characteristics of modern life. Technology can support and accompany mobility in several ways, both for business and for pleasure. Mobile phones are popular and widespread, and provide a good basis for challenging C-SPARQL use cases. In order to complete the adoption of such devices in our everyday life, mobile applications must fulfill real-time requirements, especially if we are to use them to make short-term decisions. Leveraging data from sensors, which is likely to be available in the form of streams, mobile applications may compute interesting answers by reasoning over streams.

For instance, the following C-SPARQL query finds the locations of commuters having less than 30 minutes of travel time remaining. For each user, it retrieves the train number, its position in terms of the closest station, the city where the station is in, etc., by computing the transitive closure of relation `isIn`.

```

REGISTER QUERY WhereAlmoststToDestinationCommutersAre
COMPUTED EVERY 1sec AS
SELECT DISTINCT ?user ?location
FROM <http://mobileservice.org/meansOfTransp.rdf>
FROM STREAM <http://mobileservice.org/
positions.trdf> [RANGE 10sec STEP 1sec]
WHERE {
  ?user ex:isIn ?location .
  ?user a ex:Commuter .
  ?user ex:remainingTravelTime ?t .
  FILTER ( ?t >= "PT30M"^^xsd:duration ) }

```

It does so by continuously querying an RDF stream, which describes the users on trains, moving from a station to another, together with a static RDF graph, which describes where the stations are located, e.g., a station is in a city, which is in a region, which is in a state, etc. For further information about this application scenario, the reader is directed to [6].

3.2 Oil Production

Oil operation engineers base their decision process on real time data acquired from sensors on oil rigs, located at the sea surface and seabed. A typical oil production platform is equipped with about 400.000 sensors for measuring environmental and technical parameters. Some of the problems they face include determining the expected time to failure whenever the barring starts vibrating, given the brand of the turbine, or detecting weather events from observation data.

The following C-SPARQL query detects if a weather station is observing a blizzard. A blizzard is identified when a severe storm, characterized by low temperatures, strong winds, and heavy snow, lasts for 3 hours or more.

```

REGISTER STREAM BlizzardDetection
COMPUTED EVERY 10m AS
CONSTRUCT {?s so:generatedObservation [a w:blizzard]}
FROM <http://oilprod.org/weatherStations.rdf>
FROM STREAM <http://oilprod.org/weatherObs.trdf>
[RANGE 3h STEP 10m]
WHERE {
  ?s grs:point "66.348085,10.180662" ;
  so:generatedObservation [a w:SnowfallObservation] .
  { SELECT ?s
    WHERE {
      ?s so:generatedObservation ?o1
      ?o1 a w:TemperatureObservation ;
          so:observedProperty w:AirTemperature ;
          so:result [ so:value ?temperature ] . }
    GROUP BY ( ?s )
    HAVING ( AVG(?temperature) < "0.0"^^xsd:float )
  }
  { SELECT ?s
    WHERE {
      ?s so:generatedObservation ?o2
      ?o2 a w:WindObservation ;
          so:observedProperty w:WindSpeed ;
          so:result [ so:value ?speed ] . }
    GROUP BY ( ?s )
    HAVING ( MIN(?speed) > "40.0"^^xsd:float )
  }
}

```

For further information about this application scenario, the reader is directed to [27].

4. OUTLOOK

We believe that C-SPARQL and its corresponding infrastructure provide an excellent starting point for stream reasoning [13]. By providing an RDF-based representation of heterogeneous streams, C-SPARQL solves the challenge of giving reasoners an access protocol for heterogeneous streams. As RDF is the most accepted format to feed information to reasoners, C-SPARQL allows existing reasoning mechanisms to be further extended in order to support continuous reasoning over data streams and rich background knowledge. We already made a first step in this direction, investigating the incremental maintenance of ontological entailment materializations [6]. This approach needs to be generalized to more expressive languages.

Moreover, the extraction of patterns from data streams

is subject of ongoing research in machine learning. For instance, results from statistical relational learning are able to derive classification rules from example data in very effective ways. In our future work, we intend to link relational learning methods with C-SPARQL to facilitate pattern extraction on top of RDF streams.

Finally, we envision the possibility to leverage recent developments in distributed and parallel reasoning [29, 25] as well as the compilation of reasoning tasks into queries [21] for scaling up to gigantic data streams and to extremely large numbers of concurrent reasoning tasks.

5. REFERENCES

- [1] D. J. Abadi, Y. Ahmad, M. Balazinska, U. Çetintemel, M. Cherniack, J.-H. Hwang, W. Lindner, A. S. Maskey, A. Rasin, E. Ryvkina, N. Tatbul, Y. Xing, and S. Zdonik. The Design of the Borealis Stream Processing Engine. In *Proc. Intl. Conf. on Innovative Data Systems Research (CIDR 2005)*, 2005.
- [2] A. Arasu, S. Babu, and J. Widom. The CQL Continuous Query Language: Semantic Foundations and Query Execution. *The VLDB Journal*, 15(2):121–142, 2006.
- [3] Y. Bai, H. Thakkar, H. Wang, C. Luo, and C. Zaniolo. A Data Stream Language and System Designed for Power and Extensibility. In *Proc. Intl. Conf. on Information and Knowledge Management (CIKM 2006)*, pages 337–346, 2006.
- [4] H. Balakrishnan, M. Balazinska, D. Carney, U. Çetintemel, M. Cherniack, C. Convey, E. Galvez, J. Salz, M. Stonebraker, N. Tatbul, R. Tibbetts, and S. Zdonik. Retrospective on Aurora. *The VLDB Journal*, 13(4):370–383, 2004.
- [5] S. Bandini, A. Mosca, and M. Palmonari. Common-sense spatial reasoning for information correlation in pervasive computing. *Applied Artificial Intelligence*, 21(4&5):405–425, 2007.
- [6] D. F. Barbieri, D. Braga, S. Ceri, E. Della Valle, and M. Grossniklaus. Incremental reasoning on streams and rich background knowledge. In *Proceedings of ESWC 2010, Heraklion, Crete*, 2010.
- [7] D. F. Barbieri, D. Braga, S. Ceri, and M. Grossniklaus. An Execution Environment for C-SPARQL Queries. In *Proc. Intl. Conf. on Extending Database Technology (EDBT 2010)*, 2010.
- [8] U. Bojars, J. G. Breslin, A. Finn, and S. Decker. Using the semantic web for linking and reusing data across web 2.0 communities. *Web Semantics*, 6(1):21–28, 2008.
- [9] U. Bojars, J. G. Breslin, V. Peristeras, G. Tummarello, and S. Decker. Interlinking the social web with semantics. *Intelligent Systems*, 23(3):29–40, 2008.
- [10] A. Bolles, M. Grawunder, and J. Jacobi. Streaming SPARQL – Extending SPARQL to Process Data Streams. In *Proc. Europ. Semantic Web Conf. (ESWC 2008)*, pages 448–462, 2008.
- [11] D. Brickley and R. Guha. RDF Vocabulary Description Language 1.0: RDF Schema, W3C Working Draft. Technical report, W3C, 2002.
- [12] J. Chen, D. J. DeWitt, F. Tian, and Y. Wang. NiagaraCQ: A Scalable Continuous Query System for Internet Databases. In W. Chen, J. F. Naughton, and P. A. Bernstein, editors, *Proc. ACM Intl. Conf. on Management of Data (SIGMOD 2000)*, pages 379–390, 2000.
- [13] E. Della Valle, S. Ceri, F. van Harmelen, and D. Fensel. It’s a Streaming World! Reasoning upon Rapidly Changing Information. *IEEE Intelligent Systems*, 24(6):83–89, 2009.
- [14] M. Garofalakis, J. Gehrke, and R. Rastogi. *Data Stream Management: Processing High-Speed Data Streams (Data-Centric Systems and Applications)*. Springer-Verlag New York, Inc., 2007.
- [15] L. Golab, D. DeHaan, E. D. Demaine, A. López-Ortiz, and J. I. Munro. Identifying Frequent Items in Sliding Windows over On-line Packet Streams. In *Proc. Intl. Conf. on Internet Measurement (IMC 2003)*, pages 173–178, 2003.
- [16] L. Golab and M. T. Özsu. Processing Sliding Window Multi-Joins in Continuous Queries over Data Streams. In *Proc. Intl. Conf. on Very Large Data Bases (VLDB 2006)*, pages 500–511, 2003.
- [17] Y.-N. Law, H. Wang, and C. Zaniolo. Query Languages and Data Models for Database Sequences and Data Streams. In *Proc. Intl. Conf. on Very Large Data Bases (VLDB 2004)*, pages 492–503, 2004.
- [18] Y.-N. Law and C. Zaniolo. An Adaptive Nearest Neighbor Classification Algorithm for Data Streams. In *Proc. Europ. Conf. on Principles and Practice of Knowledge Discovery in Databases (PKDD 2005)*, pages 108–120, 2005.
- [19] L. Liu, C. Pu, and W. Tang. Continual Queries for Internet Scale Event-Driven Information Delivery. *IEEE Trans. Knowl. Data Eng.*, 11(4):610–628, 1999.
- [20] M. Luther, Y. Fukazawa, M. Wagner, and S. Kurakake. Situational reasoning for task-oriented mobile service recommendation. *Knowledge Eng. Review*, 23(1):7–19, 2008.
- [21] C. Lutz, D. Toman, and F. Wolter. Conjunctive query answering in the description logic el using a relational database system. In *IJCAI’09: Proceedings of the 21st international joint conference on Artificial intelligence*, pages 2070–2075, San Francisco, CA, USA, 2009. Morgan Kaufmann Publishers Inc.
- [22] M. Mendler and S. Scheele. Exponential speedup in ul subsumption checking relative to general tboxes for the constructive semantics. In B. C. Grau, I. Horrocks, B. Motik, and U. Sattler, editors, *Description Logics*, volume 477 of *CEUR Workshop Proceedings*. CEUR-WS.org, 2009.
- [23] A. Miles, B. Matthews, M. Wilson, and D. Brickley. SKOS core: Simple Knowledge Organisation for the Web. In *Proceedings of the 2005 international conference on Dublin Core and metadata applications*, pages 1–9, Madrid, Spain, 2005. Dublin Core Metadata Initiative.
- [24] E. Prud’hommeaux and A. Seaborne. SPARQL Query Language for RDF.
- [25] A. Schlicht and H. Stuckenschmidt. Distributed resolution for expressive ontology networks. In *Web Reasoning and Rule Systems*, 2009.
- [26] D. Seid and S. Mehrotra. Grouping and aggregate queries over semantic web databases. *International Conference on Semantic Computing*, 0:775–782, 2007.
- [27] H. Stuckenschmidt, S. Ceri, E. Della Valle, and F. van Harmelen. Towards expressive stream reasoning. In *Proceedings of the Dagstuhl Seminar on Semantic Aspects of Sensor Networks*, 2010.
- [28] F. M. Suchanek, G. Kasneci, and G. Weikum. YAGO: a core of semantic knowledge unifying wordnet and wikipedia. In *WWW ’07: Proceedings of the Sixteenth International World Wide Web Conference, Banff, Canada*, pages 697–706, 2007.
- [29] J. Urbani, S. Kotoulas, J. Maassen, F. van Harmelen, and H. Bal. Owl reasoning with webpie: calculating the closure of 100 billion triples. In *Proceedings of ESWC 2010, Heraklion, Crete*, 2010.
- [30] O. Walavalkar, A. Joshi, T. Finin, and Y. Yesha. Streaming Knowledge Bases. In *Proc. Intl. Workshop on Scalable Semantic Web Knowledge Base Systems (SSWS 2008)*, 2008.